
Unbenannt
/* Programm zur Demonstration der Datenadressierung 

mit Hilfe der Adressgeneratoren
Programmentwurf: H.W.W
Datum: 04.04.2008
Tool: Visual DSP ++

*/

.SECTION /dm seg_dmda;

.VAR dmSkalar=0x55;

.VAR dmVectori[4];

.VAR dmVectorf[6]= 1.0,2e2,-0.3e-2,123.456,0.0,777.8;

.VAR dminput[]="Werte.dat";

.SECTION /pm seg_pmda;

.VAR pmSkalar;

.VAR pmVectori[4]=100,234,12345,100000000;

.VAR pmVectorf[6];

.VAR pmkopie[9];

.VAR pmkopiek[3];

.SECTION /pm seg_rth;
nop;

.align 0x4;
nop;
jump start;

.SECTION /pm seg_pmco;

/* 1.Teil
Kopieren eines Wertes von dmSkalar nach pmSkalar
Wert durch direkte Adressierung lesen und schreiben 

*/
start: R8=DM(dmSkalar); /* direkte Adressierung */

PM(pmSkalar)=R8;
/* 2.Teil

Kopieren der Werte von pmVectori nach dmVectori
1. Adressgeneratoren einstellen 
2. Werte in einer Schleife Kopieren

*/
B0=dmVectori;
M0=1;
L0=@dmVectori;
B8=pmVectori;
M8=1;
L8=@pmVectori;

loop1a: LCNTR=L8, DO loop1e UNTIL LCE;
R0=PM(I8,M8);

loop1e: DM(I0,M0)=R0;

/* 3.Teil lesen der Werte einesVektors und 
schreiben der gelesenen Werte in zwei Vektoren
Kopieren der Werte von dminput nach pmkopie und pmkopiek
1. Adressgeneratoren einstellen 
2. Werte in einer Schleife Kopieren

Die Dimension des Ziel-Vektors pmkopiek ist kleiner als die
Dimension des Quell-Vektors dminput

*/
B2=dminput;
M2=1;
L2=@dminput;
B10=pmkopie;
M10=1;
L10=@pmkopie;
B11=pmkopiek;
L11=@pmkopiek;

loop2a: LCNTR=L2, DO loop2e UNTIL LCE;
R0=DM(I2,M2);
PM(I10,M10)=R0;

loop2e: PM(I11,M10)=R0;

/* 4.Teil parallel lesen und schreiben
Kopieren der Werte von dmVectorf nach pmVectorf
1. Adressgeneratoren einstellen 
2. Werte in einer Schleife Kopieren

1. Wert vor der Schleife lesen
in der Schleife n-1-ten Wert schreiben
und parallel dazu n-ten Wert lesen
nach der Schleife n-ten Wert schreiben

*/

B1=dmVectorf;
M0=1;
L1=@dmVectorf;
B9=pmVectorf;
M9=1;
L9=@pmVectorf;
R1=@pmVectorf-1;
R0=DM(I1,M0);

loop3a: LCNTR=R1, DO loop3e UNTIL LCE;
loop3e: R0=DM(I1,M0), PM(I9,M9)=R0;

PM(I9,M9)=R0;

idle;

Seite 2


